# Lezione 8 (parte seconda)

Enrico Bertolazzi

```
> restart:
> # carico le libreri necessarie:
  # plots e plottools per disegnare i dischi
  # LinearAlgebra per i vettori e matrici
> with(plots) :
  with(plottools):
  with(LinearAlgebra) :
Warning, the name changecoords has been redefined

> # procedura gersh, data una matrice costruisce
  # la lista L che conterra` i centri e raggi di
  # Gershgorin

  gesh := proc (A)
    local L, i, j, n, C, R ;
    n := RowDimension(A) ;
    L := [] ;
    for i from 1 to n do
      C := [ Re(A[i,i]), Im(A[i,i]) ] ;
      R := add(abs(A[i,j]),j=1..i-1)+
           add(abs(A[i,j]),j=i+1..n) ;
      L := [ op(L), [ C, R ]] ;
    end ;
  end proc ;
```

$$gesh := \mathbf{proc}(A) \tag{1}$$

**local** $L, i, j, n, C, R$;

$n := LinearAlgebra\text{:-}RowDimension(A)$;

$L := [\,]$;

**for** $i$ **to** $n$ **do**

$C := [\mathrm{Re}(A[i,i]), \mathrm{Im}(A[i,i])]$;

$R := add(\mathrm{abs}(A[i,j]), j = 1..i-1) + add(\mathrm{abs}(A[i,j]), j = i+1..n)$;

$L := [op(L), [C, R]]$

**end do**

**end proc**

```
> # procedura plotgesh
  # costruisce i comandi per disegnare sia i cerchi
  # di gershgorin che la posizione degli autovalori
```

```
plotgesh := proc (A)
  local i, L, EIGS, n, LP, RMAX ;

  L  := gesh(A) ;
  n  := RowDimension(A) ;
  LP := [] ;

  # aggiunge alla lista LP gli autovalori
  EIGS := evalf(Eigenvalues(A)) ;
  RMAX := max(seq(abs(evalf(L[i][2])),i=1..n)) ;
  for i from 1 to n do
    LP := [ op(LP),
            disk([Re(EIGS[i]),Im(EIGS[i])],
            RMAX/20,color=blue) ] ;
  end ;

  # aggiunge alla lista LP i dischi
  for i from 1 to n do
    LP := [ op(LP),
            disk(L[i][1],L[i][2],color=red) ] ;
  end ;

  return LP ;
end proc ;
```

$$plotgesh := \mathbf{proc}(A) \tag{2}$$

**local** $i, L, EIGS, n, LP, RMAX$;

$L := gesh(A)$;

$n := LinearAlgebra:\text{-}RowDimension(A)$;

$LP := [\,]$;

$EIGS := evalf(LinearAlgebra:\text{-}Eigenvalues(A))$;

$RMAX := \max(seq(\mathrm{abs}(evalf(L[i][2])), i = 1..n))$;

**for** $i$ **to** $n$ **do**

$LP := [op(LP), disk([\mathrm{Re}(EIGS[i]), \mathrm{Im}(EIGS[i])], 1/20 * RMAX, color = blue)]$

**end do**;

**for** $i$ **to** $n$ **do** $LP := [op(LP), disk(L[i][1], L[i][2], color = red)]$ **end do**;

**return** $LP$

**end proc**

```
> # primo esempio

  A := <<5+3*I,1,1,1>|
        <1,-6-3*I,I,-I>|
        <1,-1-I,0,3>|
```

```
        <0,I,1,5-3*I>> ;
```

$$A := \begin{bmatrix} 5+3\,I & 1 & 1 & 0 \\ 1 & -6-3\,I & -1-I & I \\ 1 & I & 0 & 1 \\ 1 & -I & 3 & 5-3\,I \end{bmatrix} \quad (3)$$

```
> display(plotgesh(A),scaling=CONSTRAINED) ;
```




```
> # secondo esempio

  B := <<I,1/3,-1/3,1/4>|
        <1/4,-I,0,-I/4>|
        <1/4,1/4,1,1/4>|
        <0,I/4,1/4,-1>> ;
```

$$B := \begin{bmatrix} I & \frac{1}{4} & \frac{1}{4} & 0 \\ \frac{1}{3} & -I & \frac{1}{4} & \frac{1}{4}\,I \\ -\frac{1}{3} & 0 & 1 & \frac{1}{4} \\ \frac{1}{4} & -\frac{1}{4}\,I & \frac{1}{4} & -1 \end{bmatrix} \quad (4)$$

```
> display(plotgesh(B),scaling=CONSTRAINED) ;
```


1
0.5
−1.5
−1
−0.5
0
0.5
1
1.5
−0.5
−1
−1.5